NAG Library Routine Document

F03AFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03AFF computes an LU factorization of a real matrix, with partial pivoting, and evaluates the determinant.

2 Specification

```
SUBROUTINE F03AFF (N, EPS, A, LDA, D1, ID, P, IFAIL)

INTEGER            N, LDA, ID, IFAIL
REAL (KIND=nag_wp) EPS, A(LDA,*), D1, P(N)
```

3 Description

F03AFF computes an LU factorization of a real matrix A with partial pivoting: $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The determinant of A is the product of the diagonal elements of L with the correct sign determined by the row interchanges.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: EPS – REAL (KIND=nag_wp) *Input*

On entry: is no longer required by F03AFF but is retained for backwards compatibility.

3: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

On exit: A is overwritten by the lower triangular matrix L and the off-diagonal elements of the upper triangular matrix U. The unit diagonal elements of U are not stored.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03AFF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

5: D1 – REAL (KIND=nag_wp) *Output*
6: ID – INTEGER *Output*

On exit: the determinant of A is given by D1 $\times$ 2.0$^{\mathrm{ID}}$. It is given in this form to avoid overflow or underflow.

7: P(N) – REAL (KIND=nag_wp) array *Output*

On exit: P(i) gives the row index of the ith pivot.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is singular, possibly due to rounding errors. The factorization could not be completed. D1 and ID are set to zero.

IFAIL $= 2$

On entry, N < 0,
or LDA $<$ max(1, N).

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see page 107 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03AFF is approximately proportional to n^3.

9 Example

This example computes the LU factorization with partial pivoting, and calculates the determinant, of the real matrix

$$\begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix}.$$

9.1 Program Text

```
    Program f03affe

!     F03AFF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f03aff, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: d1, eps
      Integer                          :: i, id, ifail, lda, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), p(:)
!     .. Executable Statements ..
      Write (nout,*) 'F03AFF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),p(n))

      Read (nin,*)(a(i,1:n),i=1,n)

      ifail = 0
      Call f03aff(n,eps,a,lda,d1,id,p,ifail)

      Write (nout,*)
      Write (nout,*) 'Array A after factorization'

      Do i = 1, n
        Write (nout,99999) a(i,1:n)
      End Do

      Write (nout,*)
      Write (nout,*) 'Array P'
      Write (nout,99999) p(1:n)
      Write (nout,*)
      Write (nout,99998) 'D1 = ', d1, ' ID = ', id
      Write (nout,*)
      Write (nout,99998) 'Value of determinant = ', d1*2.0E0_nag_wp**id

99999 Format (1X,8F9.4)
99998 Format (1X,A,F9.4,A,I2)
    End Program f03affe
```

9.2 Program Data

```
F03AFF Example Program Data
  3
   33   16   72
  -24  -10  -57
   -8   -4  -17
```

9.3 Program Results

```
 F03AFF Example Program Results

 Array A after factorization
   -8.0000   0.5000   2.1250
  -24.0000   2.0000  -3.0000
   33.0000  -0.5000   0.3750
```

```
 Array P
    3.0000   2.0000   3.0000

 D1 =    0.3750 ID =  4

 Value of determinant =    6.0000
```